

Riproduzione di audio e video

Per riprodurre i suoni in Visual Basic sono disponibili diversi strumenti.

Per i file audio di tipo **WAV** si può utilizzare la classe **My.Computer.Audio** e i suoi metodi **Play**, per avviare l'audio, e **Stop**, per interrompere la riproduzione.

Per esempio:

```
My.Computer.Audio.Play("suono.wav", AudioPlayMode.WaitToComplete)
My.Computer.Audio.Stop( )
```

Il primo argomento del metodo *Play* definisce il nome del file, il secondo (**AudioPlayMode**) specifica la modalità di riproduzione.

WaitToComplete significa che il programma attende la fine della riproduzione audio prima di passare all'esecuzione di un'altra istruzione.

Le altre modalità sono: **Background** quando il suono viene riprodotto in sottofondo, e **BackgroundLoop** quando il suono viene ripetuto ciclicamente in sottofondo.

Il metodo **PlaySystemSound** della classe **My.Computer.Audio** riproduce invece i suoni predefiniti del sistema operativo che vengono utilizzati come avvisi all'utente.

Per esempio:

```
My.Computer.Audio.PlaySystemSound(SystemSounds.Exclamation)
```

L'enumerazione **SystemSounds** comprende anche i suoni:

SystemSounds.Asterisk

SystemSounds.Beep

SystemSounds.Hand

SystemSounds.Question.

L'uso di questi oggetti e metodi richiede la dichiarazione di importazione del namespace:

```
Imports System.Media
```

Per altri tipi di file audio, per esempio per i file **MP3**, occorre utilizzare lo specifico programma applicativo che ha generato quel tipo di audio o che comunque risulta associato ad esso come programma predefinito (per esempio *Windows Media Player*).

In questi casi si utilizza la classe **Process** che, in generale, fornisce l'accesso ai processi locali e remoti e consente di avviare e arrestare i processi del sistema. Per esempio:

```
Dim audio As Process = Process.Start("canzone1.mp3")
audio.WaitForExit( )
```

L'avvio del processo causa l'apertura della finestra del programma applicativo associato ai file di tipo MP3.

Progetto

Riprodurre un suono.

L'utente può scegliere tra due formati di file audio: WAV oppure MP3.

Nome del progetto

Audio di tipo Applicazione Windows Form.


Disegno dell'interfaccia grafica

Il form dell'interfaccia contiene un gruppo con due pulsanti di opzione (*RadioButton*) per la scelta del tipo di file audio. I tre pulsanti di comando rappresentano le funzionalità del programma: *Sfoglia* per selezionare il file da caricare, *Avvia* per iniziare la riproduzione del suono, *Stop* per interrompere. Il quarto pulsante *Esci* termina il programma.

Una casella di testo consente di visualizzare il percorso del file audio selezionato.

Il programma utilizza anche le finestre di dialogo *MessageBox* per gli avvisi all'utente.

Classe	Proprietà dell'oggetto	
Form	Name	frmAudio
	Text	Riproduzione di suoni
GroupBox	Name	grpTipo
RadioButton	Name	radio1
	Text	WAV
RadioButton	Name	radio2
	Text	MP3
TextBox	Name	txtNome
Button	Name	btnSfoglia
	Text	Sfoglia
Button	Name	btnAvvia
	Text	Avvia
Button	Name	btnStop
	Text	Stop
Button	Name	btnEsci
	Text	Esci



Gestione degli eventi

L'utente sceglie il tipo di file audio con i pulsanti di opzione. Facendo poi clic sul pulsante *Sfoglia* si apre la finestra *Apri* per selezionare il file tra quelli registrati su disco. Il filtro sui file è impostato sulla base della scelta effettuata sul tipo di file. Il nome del file selezionato viene anche scritto nella casella di testo.

Con il pulsante *Avvia* viene iniziata la riproduzione. Il pulsante *Stop* funziona solo per i file WAV perché per i file MP3 i comandi di riproduzione sono gestiti dall'applicativo associato al file audio. Il programma controlla anche che l'utente scelga un tipo di file e segnala errori di esecuzione dei file audio.

Codice Visual Basic

```
Public Class frmAudio
    Dim myAudio As String

    Private Sub btnSfoglia_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnSfoglia.Click
        Dim openFileDialog As New OpenFileDialog

        If radio1.Checked = False And radio2.Checked = False Then
            MessageBox.Show("Selezionare il tipo", "Attenzione")
        Else
            If radio1.Checked Then
                openFileDialog.Filter = "WAV Files|*.wav"
            Else
                openFileDialog.Filter = "MP3 Files|*.mp3"
            End If
            If openFileDialog.ShowDialog = Windows.Forms.DialogResult.OK Then
                myAudio = openFileDialog.FileName
                txtNome.Text = myAudio
            End If
        End If
    End Sub

    Private Sub btnAvvia_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnAvvia.Click
        If radio1.Checked = False And radio2.Checked = False Then
            MessageBox.Show("Selezionare il tipo", "Attenzione")
        Else
            If radio1.Checked Then
                Try
                    My.Computer.Audio.Play(myAudio)
                Catch ex As Exception
                    MessageBox.Show(ex.Message)
                End Try
            Else
                Try
                    Dim audio As Process = Process.Start(myAudio)
                Catch ex As Exception
                    MessageBox.Show(ex.Message)
                End Try
            End If
        End If
    End Sub
End Class
```

```

Private Sub btnStop_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnStop.Click
    If radio1.Checked Then
        My.Computer.Audio.Stop()
    End If
End Sub

Private Sub btnEsci_Click(ByVal sender As System.Object, ByVal e As
System.EventArgs) Handles btnEsci.Click
    End
End Sub

End Class

```

Per quanto riguarda la riproduzione di **file video**, la modalità è del tutto analoga a quanto visto nel progetto precedente per i file MP3.

Occorre utilizzare un'istanza della classe **Process** per avviare il programma applicativo associato ai file video.

Per esempio la riproduzione di un file **AVI** può essere realizzata con la seguente istruzione:

```
Dim video As Process = Process.Start("filmato1.avi")
```

Il metodo **Start** dell'oggetto *Process* carica il video e avvia il programma di riproduzione, per esempio *Windows Media Player*.